
TITLE: 
Falling rock catcher. 
 
PROBLEM: 
A rather rigid falling rock impacts a catcher made of an elasto-plastic net supported 
by two double-T shaped structures at the extremities. The net is 6 m and 3 m high. 
The rock has a diameter of 60 cm and a mass of 283 kg. Its initial velocity is 100 
km/h (28.3 m/s) and it impacts the catcher at the position shown above (3/4 of height, 
centred horizontally) at an angle of 45 degrees. The net cables have a diameter of 2 
mm, except the lower and higher horizontal cables which have a diameter of 1 cm. 
The thickness of the supports is 5 mm. 
  
MESH: 
Specialized 3D cable elements (FUN3) are used to discretize the net, 6720 elements 
altogether. The two supports are modelled by 640 shell elements (Q4GR). The falling 
mass is modelled by a material point. 
 
MATERIALS: 
The VM23 (Von Mises law with isotropic hardening) material is assigned to the 
supports, while the cables use the specialized FUNE material, characterized by no 
resistance in compression and an elasto-plastic behaviour in traction. Two solutions 
are presented, corresponding to rupture strains of 50% and 5%, respectively. 
 
BOUNDARY CONDITIONS: 

The two supports are clamped at the base. Contact between the falling mass and the 
net is modelled by pinballs, without taking into account any friction. 
 
LOADING: 
No external loads are assumed. 
 
INITIAL CONDITIONS 
The falling rock has an initial velocity of 100 km/h at an angle of 45 degrees towards 
the net. 
 
CALCULATION: 
The calculation is performed up to 100 ms. At the final time, the falling rock has 
completely perforated the net. 
 
POST-TREATMENT 
A detailed animation of the computed results from this calculation is available on the 
EUROPLEXUS Consortium Web site. 
 
Numerical Solutions 
 
NETCF6 
 
Solution with a cables rupture strain of 50%. The mesh generation file is: 
 
*%siz 300 
* 
opti echo 1; 
opti dime 3 elem cub8; 
opti sauv form 'netcf6.msh'; 
opti trac psc ftra 'netcf6_mesh.ps'; 
* 
p0 = 0 0 0; tol = 0.001; 
lx = 6.0; lz  = 3.0; 
nx = 40; nz = 20; 
fx = 4; fz = 4; 
dx = lx / nx; 
dz = lz / nz; 
vx = dx 0 0; vz = 0 0 dz; 
pa = (dx/2) 0 0; 
pb = (dx/2) 0 lz; 
cv = pa d (fz*nz) pb; 
pc = 0 0 (dz/2); 
pd = lx 0 (dz/2); 
ch = pc d (fx*nx) pd; 
clow = (ch plus (0 0 (0-dz/2))) coul turq; 
chig = (clow plus (0 0 lz)) coul turq; 
* 
repe loopx nx; 
  si (ega &loopx 1); 
    netv = cv; 
  sinon; 
    cv = cv plus vx; 
    netv = netv et cv; 
  finsi; 
fin loopx; 
netv = netv coul bleu; 
* 
repe loopz nz; 
  si (ega &loopz 1); 
    neth = ch; 
  sinon; 
    ch = ch plus vz; 
    neth = neth et ch; 
  finsi; 
fin loopz; 
neth = neth coul bleu; 
* 
net = netv et neth et clow et chig; 
elim tol net; 
* 
*trac qual net; 
*net1 = chan 'POI1' net; 
*trac qual net1; 
list (nbel net); 
list (nbno net); 
* 
tx = 0.1; ty = 0.2; 
ta = (0-tx/2) (0-ty/2) 0; 
tb = (  tx/2) (0-ty/2) 0; 
tc = 0        (0-ty/2) 0; 
td = (0-tx/2) (  ty/2) 0; 
te = (  tx/2) (  ty/2) 0; 
tf = 0        (  ty/2) 0; 
ntx = 1; nty = 2; 

t1 = ta d ntx tc d ntx tb; 
t2 = td d ntx tf d ntx te; 
t3 = tc d nty p0 d nty tf; 
tt = t1 et t2 et t3; 
elim tol tt; 
supp1 = (tt tran (2*nz) (0 0 lz)) coul vert; 
supp2 = (supp1 plus (lx 0 0)) coul vert; 
supp = supp1 et supp2; 
list (nbel supp); 
list (nbno supp); 
stru = supp et net; 
elim tol stru; 
*trac cach face stru; 
list (nbel stru); 
list (nbno stru); 
* 
bloq = supp poin plan p0 (1 0 0) (0 1 0) tol; 
list (nbno bloq); 
*trac bloq; 
* 
rfall = 0.3; 
pfall = (lx/2) (rfall) (3*lz/4); 
list pfall; 
fall = (manu 'POI1' pfall) coul roug; 
*trac cach face (stru et fall); 
* 
fac = 1.5; 
*rfall2 = fac*rfall*rfall; 
rfall2 = fac*rfall; 
xf yf zf = coor pfall; 
n = nbel net; 
ni = 0; 
repe loop3 n; 
  ei = net elem &loop3; 
  xb yb zb = coor (bary ei); 
  dx = xb - xf; 
* dz = zb - zf; 
* d2 = (dx*dx) + (dz*dz); 
  si (abs(dx) < rfall2); 
* si (d2 < rfall2); 
    ni = ni + 1; 
    si (ega ni 1); 
      neti = ei; 
    sinon; 
      neti = neti et ei; 
    finsi; 
  finsi; 
fin loop3; 
*neti = neti coul roug; 
list ni; 
*trac qual neti; 
neto = diff net neti; 
*trac qual neto; 
trac face cach (neti et neto et supp et fall); 
* 
mesh = supp et neti et neto et fall et bloq; 
tass mesh; 
sauv form mesh; 
* 
fin; 

 

The input file is: 
 
NETCF6 
*----------------------------------------------------------------------- 
ECHO 
 conv win 
CAST mesh 
*-----------------------------------------------------------Problem type 
TRID NONL LAGR FAIL 
*-----------------------------------------------------------Dimensioning 
DIME 
   PT6L 738 PT3L 5859 FUN3 6720 PMAT 1 Q4GR 640 ZONE 3 
TERM 
*---------------------------------------------------------------Geometry 

GEOM FUN3 net PMAT fall Q4GR supp TERM 
*--------------------------------------------------Geometric Complements 
COMP EPAI 3.14E-6 LECT netv neth TERM ! 2 mm diameter 
     EPAI 7.85E-5 LECT clow chig TERM ! 1 cm diameter  
     EPAI 0.6     LECT fall TERM ! only for graphical representation 
     EPAI 0.005   LECT supp TERM      ! 5 mm thickness 
*----------------------------------------------------------Material data 
MATE FUNE RO 8000. YOUN 2.0E11 NU 0.3 ERUP 0.5 
          ELAS 0.440D9 
          TRAC 5 0.440D9 2.200D-3 0.735D9 0.156D0 
                 0.900D9 0.283D0 1.077D9 0.475D0 1.142D9 0.600D0 
          LECT net TERM 

     MASS 282.75 LECT fall TERM 
     VM23 RO 8000. YOUN 1.95D11 NU 0.3 ELAS 1.95E+8 
          TRAC 4 1.95E+8 .001 2.16E+8 .005 2.6E+8 .02 2.6E8 10. 
          LECT supp TERM 
*----------------------------------------------------Boundary conditions 
LINK COUP BLOQ 123456 LECT bloq TERM 
     PINB BODY MLEV 0   LECT neti TERM 
          BODY DIAM 0.6 LECT fall TERM 
*-----------------------------------------------------Initial conditions 
INIT VITE 2 -20 LECT fall TERM ! v_0 = about 100 km/h 
     VITE 3 -20 LECT fall TERM 
*----------------------------------------------------------------Outputs 
ECRI VITE TFRE 100.E-3 
     FICH ALIC      TFRE 1.E-3 
*----------------------------------------------------------------Options 
OPTI NOTE 
     LOG 1 
*--------------------------------------------------Transient calculation 
CALC TINI 0 TFIN 0.100 
*==============================================================ANIMATION 
PLAY 
CAME   1 EYE   8.41798E+00 -6.59217E+00  1.20611E+01 
!        Q     8.68252E-01  3.25861E-01  9.16021E-02  3.62715E-01 
         VIEW -3.95456E-01  4.99408E-01 -7.70848E-01 
         RIGH  7.20094E-01  6.89555E-01  7.73214E-02 
         UP   -5.70157E-01  5.24506E-01  6.32309E-01 
         FOV   2.48819E+01 
!CAME   2 EYE   4.76084E+00 -1.97370E+00  4.93234E+00 
!!        Q     8.68252E-01  3.25861E-01  9.16021E-02  3.62715E-01 
!         VIEW -3.95456E-01  4.99408E-01 -7.70848E-01 
!         RIGH  7.20094E-01  6.89555E-01  7.73214E-02 
!         UP   -5.70157E-01  5.24506E-01  6.32309E-01 
!         FOV   2.48819E+01 
CAME   2 EYE   8.34687E+00 -4.82530E+00  4.41693E+00 
!        Q     7.83098E-01  4.84674E-01  2.27182E-01  3.16603E-01 
         VIEW -6.62709E-01  6.15242E-01 -4.26958E-01 
         RIGH  6.96303E-01  7.16080E-01 -4.89126E-02 
         UP   -2.75643E-01  3.29707E-01  9.02948E-01 
         FOV   2.48819E+01 
 
!Sequence 01: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit1 'EUROPLEXUS (C) Animation' 
     tit2 'Falling Rock Catcher' 
     tit3 'Author: F. Casadei' 
sler cam1 1        nfra 30 
trac offs fich avi      nocl nfto 905 fps 10 kfre 10 comp -1 rend 
 
!Sequence 02: Title (30 frames) 
!------------------------------ 
titl tit2 'Geometry and Pinballs' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl                                 rend 
 
!Sequence 03: Still from cam1 (15 frames) 
!---------------------------------------- 
scen geom navi free 
          line sfre !heou 
          poin sphp 
         !pinb pare cdes 
     colo pape 
     lima on 
sler cam1 1        nfra 15 
trac offs fich avi cont nocl                                 rend 
 
!Sequence 04: Still from cam1 with pinballs (15 frames) 
!------------------------------------------------------ 
scen geom navi free 
          line sfre !heou 
         !poin sphp 
          pinb pare cdes 
     colo pape 
     lima on 
sler cam1 1        nfra 15 
trac offs fich avi cont nocl                                 rend 
 
!Sequence 05: Slerping from cam1 to cam2 (30 frames) 
!--------------------------------------------------- 
scen geom navi free 
          line sfre !heou 
         !poin sphp 
          pinb pare cdes 
     colo pape 
     lima on 
sler cam1 1 cam2 2 nfra 30 
trac offs fich avi cont nocl                                 rend 
 
!Sequence 06: Transient (100 frames) 
!----------------------------------- 
sler cam1 2        nfra 1 
freq 0 tfre 1.e-3 
gotr loop 99 offs fich avi cont nocl obje nfai lect tous term rend 
go 
trac offs fich avi cont nocl obje nfai lect tous term              rend 
 
ENDPLAY 
*=========================================================POST-TREATMENT 
SUIT 
NETCF6 POST 
*----------------------------------------------------------------------- 
ECHO 
 conv win 
* 
RESU ALIC GARD PSCR 
* 
OPTI PRIN 
* 
SORT VISU NSTO 1 
*==============================================================ANIMATION 
PLAY 
CAME   1 EYE   8.41798E+00 -6.59217E+00  1.20611E+01 
!        Q     8.68252E-01  3.25861E-01  9.16021E-02  3.62715E-01 
         VIEW -3.95456E-01  4.99408E-01 -7.70848E-01 
         RIGH  7.20094E-01  6.89555E-01  7.73214E-02 
         UP   -5.70157E-01  5.24506E-01  6.32309E-01 
         FOV   2.48819E+01 
CAME   2 EYE   8.34687E+00 -4.82530E+00  4.41693E+00 
!        Q     7.83098E-01  4.84674E-01  2.27182E-01  3.16603E-01 
         VIEW -6.62709E-01  6.15242E-01 -4.26958E-01 
         RIGH  6.96303E-01  7.16080E-01 -4.89126E-02 
         UP   -2.75643E-01  3.29707E-01  9.02948E-01 
         FOV   2.48819E+01 
CAME   3 EYE   1.19987E+00 -1.01152E+00  8.34224E-01 
!        Q     7.57665E-01  5.12725E-01  2.51753E-01  3.15716E-01 
         VIEW -7.05240E-01  6.17983E-01 -3.47466E-01 
         RIGH  6.73888E-01  7.36574E-01 -5.77375E-02 
         UP   -2.20254E-01  2.74872E-01  9.35913E-01 
         FOV   2.48819E+01 
 
!Sequence 07: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit2 'Geometry only' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl                                 rend 
 
!Sequence 08: Transient from cam1 (101 frames) 
!--------------------------------------------- 
scen geom navi free 
          line sfre heou 
          face sbac 
          poin sphp 
         !pinb pare cdes 
     colo pape 
     lima on 
sler cam1 1        nfra 1 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
gotr loop 99 offs fich avi cont nocl                        
     obje nfai lect tous term rend 

go 
trac offs fich avi cont nocl                                      
     obje nfai lect tous term rend 
ENDPLAY 
*======================================================================= 
     VISU NSTO 1 
*======================================================================= 
PLAY 
 
!Sequence 09: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit2 'Velocities' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl rend 
 
!Sequence 10: Transient from cam1 (101 frames) 
!--------------------------------------------- 
scen geom navi free 
          line sfre heou 
          face sbac 
          poin sphp 
         !pinb pare cdes 
     colo pape 
     lima on 
     VECT SCCO FIEL VITE SCAL USER PROG 2 PAS 2 28 TERM 
     text vsca 
sler cam1 1        nfra 1 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
gotr loop 99 offs fich avi cont nocl                        
     obje nfai lect tous term rend 
go 
trac offs fich avi cont nocl                                      
     obje nfai lect tous term rend 
ENDPLAY 
*======================================================================= 
     VISU NSTO 1 
*======================================================================= 
PLAY 
 
!Sequence 11: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit2 'Strain in Cables' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl rend 
 
!Sequence 12: Transient from cam1 (101 frames) 
!--------------------------------------------- 
scen geom navi free 
          line sfre heou 
          face sbac 
          poin sphp 
         !pinb pare cdes 
     colo pape 
     lima on 
     ISO  FILL FIEL ECRO 1 SCAL USER PROG 1.E-2 PAS 1.E-2 14.E-2 TERM 
               SUPP LECT net TERM 
     text isca 
sler cam1 1        nfra 1 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
gotr loop 99 offs fich avi cont nocl                        
     obje nfai lect tous term rend 
go 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
ENDPLAY 
*======================================================================= 
     VISU NSTO 1 
*======================================================================= 
PLAY 
 
!Sequence 13: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit2 'Hardening in Supports' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl rend 
 
!Sequence 14: Transient from cam1 (101 frames) 
!--------------------------------------------- 
scen geom navi free 
          line sfre heou 
          face sbac 
          poin sphp 
         !pinb pare cdes 
     colo pape 
     lima on 
     ISO  FILL FIEL ECRO 7 SCAL USER PROG 1.95E8 PAS 0.05E8 2.6E8 TERM 
               SUPP LECT supp TERM 
     text isca 
sler cam1 1        nfra 1 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
gotr loop 99 offs fich avi cont nocl                        
     obje nfai lect tous term rend 
go 
trac offs fich avi cont nocl 
     obje nfai lect tous term rend 
ENDPLAY 
*======================================================================= 
     VISU NSTO 1 
*======================================================================= 
PLAY 
 
!Sequence 15: Title (30 frames) 
!------------------------------ 
scen geom navi free 
     colo pape 
titl tit2 'Zoom on Support 1' 
sler cam1 1        nfra 30 
trac offs fich avi cont nocl rend 
 
!Sequence 16: Slerp from cam1 to cam3 (30 frames) 
!------------------------------------------------ 
scen geom navi free 
          line heou sfre 
          face sbac 
          poin sphp 
         !pinb cdes 
     colo pape 
     lima on 
sler cam1 1 cam2 3 nfra 30 
trac offs fich avi cont nocl rend 
 
!Sequence 17: Transient from cam3 (101 frames) 
!--------------------------------------------- 
scen geom navi free 
          line heou sfre 
          face sbac 
          poin sphp 
         !pinb cdes 
     colo pape 
     lima on 
     ISO  FILL FIEL ECRO 7 SCAL USER PROG 1.95E8 PAS 0.05E8 2.6E8 TERM 
               SUPP LECT supp1 TERM 
     text isca 
sler cam1 3        nfra 1 
trac offs fich avi cont nocl 
     obje lect supp1 net term rend 
gotr loop 99 offs fich avi cont nocl                        
     obje lect supp1 net term rend 
go 
trac offs fich avi cont 
     obje lect supp1 net term rend 
ENDPLAY 
*======================================================================= 
FIN 

The initial configuration with pinballs shown is: 
 

 
 
Rupture of the cables starts at around 75 ms and is completed at about 86 ms: 

The final configuration (geometry only) and the final velocities are: 
 

 
 
The strain in the cables at 86 ms and the hardening in the suuports is: 
 

 
 
 
 
 
 
 
 
 
 
 

NETCF8 
 
Solution with a cables rupture strain of 5%. Rupture of the cables starts at around 35 
ms and is completed at about 53 ms: 

The final configuration (geometry only) and the final velocities are: 
 

 
 
The strain in the cables at 53 ms and the hardening in the supports are: 
 

 


